
Programación declarativa (13–Febrero–2004)

Apellidos:

Nombre:

Ejercicio 1 [2 puntos] Definir la relación multiplicada(+L1,+N,-L2) que se verifica si L2
es la lista obtenida repitiendo N veces los elementos de la lista L1. Por ejemplo,

?- multiplicada([a,b,c],3,L).

L = [a, a, a, b, b, b, c, c, c]

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
Solución:

multiplicada(L1,N,L2) :-

multiplicada_aux(L1,N,N,L2).

donde multiplicada_aux(+L1,+K,+N,-L2) se verifica si L2 es la lista obtenida repitiendo
el primer elemento de L1 K veces y los restantes N veces. Por ejemplo,

?- multiplicada_aux([a,b,c],2,3,L).

L = [a, a, b, b, b, c, c, c]

Su definición es

multiplicada_aux([],_,_,[]).

multiplicada_aux([_|L1],0,N,L2) :-

multiplicada_aux(L1,N,N,L2).

multiplicada_aux([X|L1],K,N,[X|L2]) :-

K > 0,

K1 is K-1,

multiplicada_aux([X|L1],K1,N,L2).

Ejercicio 2 [1.5 puntos] Definir la relación operación_listas(+O,+L1,+L2,-L3) que se
verifica si L3 es la lista obtenida aplicando la operación binaria O a los elementos de L1 y L2

que ocupan la misma posición. Por ejemplo,

?- operación_lista(+,[1,2,3],[4,5,6],L).

L = [5, 7, 9]

?- operación_lista(*,[1,2,3],[4,5,6],L).

L = [4, 10, 18]

(Nota: Se supone que L1 y L2 tienen la misma longitud)
. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
Solución:

operación_lista(_,[],[],[]).

operación_lista(O,[X1|L1],[X2|L2],[X3|L3]) :-

E =.. [O,X1,X2],

X3 is E,

operación_lista(O,L1,L2,L3).



Ejercicio 3 [1.5 puntos] Definir la relación subconjunto_maximal(+L1,-L2) que se verifica
si L2 es un subconjunto maximal de L1 (es decir, es un conjunto de elementos de L1 tal que
sólo existe un elemento de L1 que no pertenece a L2). Por ejemplo,

?- subconjunto_maximal([c,b,a,b,c,a,c],L).

L = [b, c] ; L = [a, c] ; L = [a, b] ;

No

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
Solución:

subconjunto_maximal(L1,L2) :-

setof(X,member(X,L1),L3),

select(_,L3,L2).

Ejercicio 4 [2 puntos] Definir la relación codificación(+L1,-L2) que se verifica si L2 es el
conjunto formado por los pares de la forma X-N donde X es un elemento de L1 y N el número
de ocurrencias de X en L1. Por ejemplo,

?- codificación([a,a,b,a,b,b,c,b],L).

L = [a-3, b-4, c-1]

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
Solución:

codificación([],[]).

codificación([X|L1],[X-N|L2]) :-

cuenta_y_resto(X,[X|L1],N,L3),

codificación(L3,L2).

donde cuenta_y_resto(+X,+L1,-N,-L2) se verifica si N es el número de veces que ocurre X

en la lista L1 y L2 es la lista obtenida de L1 eliminando las ocurrencias de X. Por ejemplo,

?- cuenta_y_resto(a,[a,b,a,c,a,a,d],N,L).

N = 4

L = [b, c, d]

Su definición es

cuenta_y_resto(_,[],0,[]).

cuenta_y_resto(X,[X|L1],N,L2) :-

!,

cuenta_y_resto(X,L1,N1,L2),

N is N1+1.

cuenta_y_resto(X,[Y|L1],N,[Y|L2]) :-

% X \= Y

cuenta_y_resto(X,L1,N,L2).

Segunda parte [3 puntos] Examen práctico en el laboratorio.


